Needles in a Haystack: Special Varieties via Small Fields

Frank-Olaf Schreyer and Fabio Tonoli

In this article we illustrate how picking points over a finite field at random can help to investigate algebraic geometry questions. In the first part we develop a program that produces random curves of genus $g \leq 14$. In the second part we use the program to test Green's Conjecture on syzygies of canonical curves and compare it with the corresponding statement for Coble self-dual sets of points. In the third section we apply our techniques to produce Calabi-Yau 3-folds of degree 17 in $\mathbb{P}^6$.

Introduction

The advances in speed of modern computers and computer algebra systems gave life to the idea of solving equations by guessing a solution. Suppose $\mathbb{M} \subset \mathbb{G}$ is a subvariety of a rational variety of codimension c. Then we expect that the probability for a point $p \in \mathbb{G}(\mathbb{F}_q)$ to lie in $\mathbb{M}(\mathbb{F}_q)$ is about $1/q^c$. Here $\mathbb{F}_q$ denotes the field with q elements.

We will discuss this idea in the following setting: $\mathbb{M}$ will be a parameter space for objects in algebraic geometry, e.g., a Hilbert scheme, a moduli space, or a space dominating such spaces.

The most basic question we might have in this case is whether $\mathbb{M}$ is non-empty and whether an open part of $\mathbb{M}$ corresponds to smooth objects.

Typically in these cases we will not have explicit equations for $\mathbb{M} \subset \mathbb{G}$ but only an implicit algebraic description of $\mathbb{M}$, and our approach will be successful if the time required to check $p \notin \mathbb{M}(\mathbb{F}_q)$ is sufficiently small compared to q^c. The first author applied this method first in [32] to construct some rational surfaces in $\mathbb{P}^4$; see [15,11] for motivation.

In this first section we describe a program that picks curve of genus $g \leq 14$ at random. The moduli spaces $\mathfrak{M}_g$ are known to be unirational for $g \leq 13$; see [33,8].

Our approach based on this result can viewed as a computer aided proof of the unirationality. Many people might object that this not a proof because we cannot control every single step in the computation. We however think that such a proof is much more reliable than a proof based on man-made computations. A mistake in a computer aided approach most often leads to an output far away from our expectation, hence it is easy to spot. A substantial improvement of present computers and computer algebra systems would give us an explicit unirational parametrization of $\mathfrak{M}_g$ for $g \leq 13$.

In the second part we apply our "random curves" to probe the consequences of Green's conjecture on syzygies of canonical curves, and compare these results with the corresponding statements for "Coble self-dual" sets of $2g-2$ points in $\mathbb{P}^{g-2}$.

In the last section we exploit our method to prove the existence of three components of the Hilbert scheme of Calabi-Yau 3-folds of degree 17 in $\mathbb{P}^6$ over the complex numbers. This is one of the main results of the second author's thesis [34, Chapter 4]. Calabi-Yau threefolds of lower degree in $\mathbb{P}^6$ are easy to construct, using the Pfaffian construction and a study of their Hartshorne-Rao modules. For degree 17 the Hartshorne-Rao module has to satisfy a subtle condition. Explicit examples of such Calabi-Yau 3-folds are first constructed over a finite field by our probabilistic method. Then a delicate semi-continuity argument gives us the existence of such Calabi-Yau 3-folds over some number field.

Acknowledgments. We thank Hans-Christian v. Bothmer and Dan Grayson for valuable discussions and remarks.

Notation. For a finitely generated graded module M over the polynomial ring $S = k[x_0, \dots, x_r]$ we summarize the numerical information of a finite free resolution

$$0 \leftarrow M \leftarrow F_0 \leftarrow F_1 \leftarrow \dots \leftarrow F_n \leftarrow 0$$

with $F_i = \oplus_j S(-j)^{\beta_{ij}}$ in a table of Betti numbers, whose ij^{th} entry is

$$\beta_{i,i+j} = \dim \operatorname{Tor}_i^S(M,k)_{i+j}.$$

As in the *Macaulay 2* command `betti` we suppress zeroes. For example the syzygies of the rational normal curve in $\mathbb{P}^3$ have the following Betti table.

1	-	-
-	3	2

Note that the degrees of the entries of the matrices in the free resolution can be read off from the relative position of two numbers in consecutive columns. A pair of numbers in a line corresponds to linear entries. Quadratic entries correspond to two numbers of a square. Thus

1	-	-	-
-	5	5	-
-	-	-	1

corresponds to a 4 term complex with a quadratic, a linear and another quadratic map. The Grassmannian $\mathbb{G}(2,5)$ in its Plücker embedding has such a free resolution.

1 How to Make Random Curves up to Genus 14

The moduli space of curves $\mathfrak{M}_g$ is known to be of general type for $g \geq 24$ and has non-negative Kodaira dimension for $g = 23$ by work of Harris, Mumford and Eisenbud [21,13]. For genus $g \leq 13$ unirationality is known [8,33]. In this section we present a *Macaulay 2* program that over a finite field $\mathbb{F}_q$ picks a point in $\mathfrak{M}_g(\mathbb{F}_q)$ for $g \leq 14$ at random.

By Brill-Noether theory [2] every curve of genus g has a linear system g^r_d of dimension r and degree d, provided that the Brill-Noether number ρ satisfies

$$\rho := \rho(g, d, r) := g - (r + 1)(g - d + r) \geq 0.$$

We utilize this to find appropriate (birational) models for general curves of genus g.

1.1 Plane Models, $g \leq 10$

This case was known to Severi; see [1]. Choose $d = g + 2 - \lfloor g/3 \rfloor$. Then $\rho(g, d, 2) \geq 0$ i.e., a general curve of genus g has a plane model C' of degree d. We expect that C' has

$$\delta = \binom{d-1}{2} - g$$

double points. If the double points are in general position, then

$$s = h^0(\mathbb{P}^2, \mathcal{O}(d)) - 3\delta - 1$$

is the expected dimension of the linear system of curves of degree d with δ assigned double points. We have the following table:

g	1	2	3	4	5	6	7	8	9	10	11	12
ρ	1	2	0	1	2	0	1	2	0	1	2	0
d	3	4	4	5	6	6	7	8	8	9	10	10
δ	0	1	0	2	5	4	8	13	12	18	25	24
s	9	11	14	14	12	15	11	5	8	0	-10	-7

Thus for $g \leq 10$ we assume that these double points lie in general position. For $g > 10$ the double points cannot lie in general position because $s < 0$. Since it is difficult to describe the special locus $H_\delta(g) \subset \mathrm{Hilb}_\delta(\mathbb{P}^2)$ of double points of nodal genus g curves, the plane model approach collapses for $g > 10$.

Random Points. In our program, which picks plane models at random from an Zariski open subspace of $\mathfrak{M}_g$, we start by picking the nodes. However, over a small field $\mathbb{F}_q$ it is not a good idea to pick points individually, because there might be simply too few: $|\mathbb{P}^2(\mathbb{F}_q)| = 1+q+q^2$. What we should do is to pick a collection Γ of δ points in $\mathbb{P}^2(\bar{\mathbb{F}}_q)$ that is defined over $\mathbb{F}_q$. General points in $\mathbb{P}^2$ satisfy the minimal resolution condition, that is, they have expected Betti numbers. This follows from the Hilbert-Burch theorem [12, Theorem 20.15]. If the ideal of such Γ has generators in minimal degree k, then $\binom{k+1}{2} \leq \delta < \binom{k+2}{2}$, which gives $\delta = \binom{k+1}{2} + \epsilon$ with $0 \leq \epsilon \leq k$. Thus $k = \lceil(-3+\sqrt{9+8\delta})/2\rceil$. The Betti table is one of the following two tables:

$$2\epsilon \leq k : \quad \begin{array}{r|lll} 0 & 1 & - & - \\ 1 & - & - & - \\ \vdots & \vdots & \vdots & \vdots \\ k-2 & - & - & - \\ k-1 & - & k+1-\epsilon & k-2\epsilon \\ k & - & - & \epsilon \end{array} \qquad 2\epsilon \geq k : \quad \begin{array}{r|lll} 0 & 1 & - & - \\ 1 & - & - & - \\ \vdots & \vdots & \vdots & \vdots \\ k-2 & - & - & - \\ k-1 & - & k+1-\epsilon & - \\ k & - & 2\epsilon-k & \epsilon \end{array}$$

So we can specify a collection Γ of δ points by picking the Hilbert-Burch matrix of their resolution; see [12, Thm 20.15]. This is a matrix with linear and quadratic entries only, whose minors of size ϵ ($k-\epsilon$ if $2\epsilon \leq k$) generate the homogeneous ideal of Γ.

```
i1 : randomPlanePoints = (delta,R) -> (
          k:=ceiling((-3+sqrt(9.0+8*delta))/2);
          eps:=delta-binomial(k+1,2);
          if k-2*eps>=0
          then minors(k-eps,
               random(R^(k+1-eps),R^{k-2*eps:-1,eps:-2}))
          else minors(eps,
               random(R^{k+1-eps:0,2*eps-k:-1},R^{eps:-2})));
```

In unlucky cases these points might be infinitesimally near.

```
i2 : distinctPoints = (J) -> (
          singJ:=minors(2,jacobian J)+J;
          codim singJ == 3);
```

The procedure that returns the ideal of a random nodal curve is then straightforward:

```
i3 : randomNodalCurve = method();

i4 : randomNodalCurve (ZZ,ZZ,Ring) := (d,g,R) -> (
          delta:=binomial(d-1,2)-g;
          K:=coefficientRing R;
          if (delta==0)
          then (     --no double points
               ideal random(R^1,R^{-d}))
          else (      --delta double points
               Ip:=randomPlanePoints(delta,R);
               --choose the curve
               Ip2:=saturate Ip^2;
               ideal (gens Ip2 * random(source gens Ip2, R^{-d}))));

i5 : isNodalCurve = (I) -> (
          singI:=ideal jacobian I +I;delta:=degree singI;
          d:=degree I;g:=binomial(d-1,2)-delta;
          {distinctPoints(singI),delta,g});
```

We next ask if we indeed get in this way points in a parameter space that dominates $\mathfrak{M}_g$ for $g \leq 10$. Let $\mathrm{Hilb}_{(d,g)}(\mathbb{P}^2)$ denote the Hilbert scheme of nodal plane curves of degree d and geometric genus g. Our construction starts from a random element in $\mathrm{Hilb}_\delta(\mathbb{P}^2)$ and picks a random curve in the corresponding fiber of $\mathrm{Hilb}_{(d,g)}(\mathbb{P}^2) \to \mathrm{Hilb}_\delta(\mathbb{P}^2)$:

$$\begin{array}{ccc} \mathrm{Hilb}_{(d,g)}(\mathbb{P}^2) & \longrightarrow & \mathfrak{M}_g \ . \\ \downarrow & & \\ \mathrm{Hilb}_\delta(\mathbb{P}^2) & & \end{array}$$

So the question is whether $\mathrm{Hilb}_{(d,g)}(\mathbb{P}^2)$ dominates $\mathrm{Hilb}_\delta(\mathbb{P}^2)$. A naive dimension count suggests that this should be true: the dimension of our parameter space is given by $2\delta + s$, which is $3(g-1) + \rho + \dim \mathrm{PGL}(3)$, as it should be. To conclude this there is more to verify: it could be that the nodal models of general curves have double points in special position, while all curve constructed above lie over a subvariety of $\mathfrak{M}_g$. One way to exclude this is to prove that the variety $G(g,d,2)$ over $\mathfrak{M}_g$, whose fiber over a curve $\tilde{C} \in \mathfrak{M}_g$ is $G^2_d(\tilde{C}) = \{g^2_d\text{'s}\}$, is irreducible or, to put it differently, that the Severi Conjecture holds:

Theorem 1.1 (Harris [20]). *The space of nodal degree d genus g curves in $\mathbb{P}^2$ is irreducible.*

Another much easier proof for the few (d,g) we are interested in is to establish that our parameter space $\mathbb{M}$ of the construction is smooth of expected dimension at our random point $p \in \mathbb{M}$, as in [1]. Consider the following diagram:

$$\mathbb{H} = \mathrm{Hilb}_{(d,g)}(\mathbb{P}^2)/\operatorname{Aut}(\mathbb{P}^2) \xrightarrow{\pi} \mathfrak{M}_g.$$

For a given curve $\tilde{C} \in \mathfrak{M}_g$, the inverse image $\pi^{-1}(\tilde{C})$ consists of the variety $W^2_d(\tilde{C}) \subset \mathrm{Pic}^d(\tilde{C})$. Moreover the choice of a divisor $L \in W^2_d(\tilde{C})$ is equivalent to the choice of $p \in \mathbb{M}$, modulo $\operatorname{Aut}(\mathbb{P}^2)$: indeed p determines a morphism $\nu\colon \tilde{C} \longrightarrow C \subset \mathbb{P}^2$ and a line bundle $L = \nu^{-1}\mathcal{O}_{\mathbb{P}^2}(1)$, where $\tilde{C}$ is the normalization of the (nodal) curve C. Therefore $\mathbb{M}$ is smooth of expected dimension $3(g-1)+\rho+\dim \mathrm{PGL}(3)$ at $p \in \mathbb{M}$ if and only if $W^2_d(\tilde{C})$ is smooth of expected dimension ρ in L. This is well known to be equivalent to the injectivity of the multiplication map μ_L

$$H^0(L) \otimes H^0(K_{\tilde{C}} \otimes L^{-1}) \xrightarrow{\mu_L} H^0(K_{\tilde{C}}),$$

which can be easily checked in our cases, see [2, p. 189]. In our cases μ_L can be rewritten as

$$H^0(\mathcal{O}_{\mathbb{P}^2}(1)) \otimes H^0(I_\Gamma(d-4)) \xrightarrow{\mu_L} H^0(I_\Gamma(d-3)).$$

So we need two conditions:

(1) $H^0(I_\Gamma(d-5)) = 0$;
(2) there are no linear relations among the generators of $H^0(I_\Gamma(d-4))$ of degree $d-3$.

We proceed case by case. For genus $g \leq 5$ this is clear, since $H^0(I_\Gamma(d-4)) = 0$ for $g = 2, 3$ and $\dim H^0(I_\Gamma(d-3)) = 1$ for $g = 4, 5$. For $g = 6$ we have $\dim H^0(I_\Gamma(d-3)) = \dim H^0(I_\Gamma(2)) = 2$ and the Betti numbers of Γ

1	-	-
-	2	-
-	-	1

shows there are no relations with linear coefficients in $H^0(I_\Gamma(2))$. For $7 \leq g \leq 10$ the method is similar: everything is clear once the Betti table of resolution of the set of nodal points Γ is computed. As a further example we do here the case $g = 10$: we see that $\dim H^0(I_\Gamma(d-3)) = \dim H^0(I_\Gamma(5)) = 3$ and the Betti numbers of Γ are

1	-	-
-	-	-
-	-	-
-	-	-
-	3	-
-	1	3

from which it is clear that there are no linear relations between the quintic generators of I_Γ.

1.2 Space Models and Hartshorne-Rao Modules

The Case of Genus $g = 11$. In this case we have $\rho(11, 12, 3) = 3$. Hence every general curve of genus 11 has a space model of degree 12. Moreover for a general curve the general space model of this degree is linearly normal, because $\rho(11, 13, 4) = -1$ takes a smaller value. If moreover such a curve $C \subset \mathbb{P}^3$ has maximal rank, i.e., for each $m \in \mathbb{Z}$ the map

$$H^0(\mathbb{P}^3, \mathcal{O}(m)) \to H^0(C, \mathcal{O}_C(m))$$

has maximal rank, then the Hartshorne-Rao module M, defined as $M = H^1_*(\mathcal{I}_C) = \oplus_m H^1(\mathbb{P}^3, \mathcal{I}_C(m))$, has Hilbert function with values $(0, 0, 4, 6, 3, 0, 0, \dots)$. For readers who want to know more about the Hartshorne-Rao module, we refer to the pleasant treatment in [24].

Since being of maximal rank is an open condition, we will try a construction of maximal rank curves. Consider the vector bundle $\mathcal{G}$ on $\mathbb{P}^3$ associated to the first syzygy module of I_C:

$$0 \leftarrow \mathcal{I}_C \leftarrow \oplus_i \mathcal{O}(-a_i) \leftarrow \mathcal{G} \leftarrow 0 \qquad (1)$$

In this set-up $H^2_*(\mathcal{G}) = H^1_*(\mathcal{I}_C)$. Thus $\mathcal{G}$ is, up to direct sum of line bundles, the sheafified second syzygy module of M; see e.g., [10, Prop. 1.5].

The expected Betti numbers of M are

4	10	3	-	-
-	-	8	2	-
-	-	-	6	3

Thus the $\mathbb{F}$-dual $M^* = \mathrm{Hom}_{\mathbb{F}}(M, \mathbb{F})$ is presented as $\mathbb{F}[x_0, \ldots, x_3]$-module by a 3×8 matrix with linear and quadratic entries, and a general such matrix will give a general module (if the construction works, i.e., if the desired space of modules is non-empty), because all conditions we impose are semi-continuous and open. Thus M depends on

$$\dim \mathbb{G}(6, 3h^0\mathcal{O}(1)) + \dim \mathbb{G}(2, 3h^0\mathcal{O}(2) - 6h^0\mathcal{O}(1)) - \dim SL(3) = 36$$

parameters.

Assuming that C has minimal possible syzygies:

1	-	-	-
-	-	-	-
-	-	-	-
-	-	-	-
-	6	2	-
-	-	6	3

we obtain, by dualizing the sequence (1), the following exact sequence

$$\mathcal{G}^* \leftarrow 6\mathcal{O}(5) \leftarrow \mathcal{O} \leftarrow 0.$$

If everything is as expected, i.e., the general curve is of maximal rank and its syzygies have minimal possible Betti numbers, then the entries of the right hand matrix are homogeneous polynomials that generate I_C. We will compute I_C by determining $\ker(\phi\colon 6\mathcal{O}(5) \to \mathcal{G}^*)$. Comparing with the syzygies of M we obtain the following isomorphism

$$\mathcal{G}^* \cong \ker(2\mathcal{O}(6)\oplus 6\mathcal{O}(7) \to 3\mathcal{O}(8)) \cong \mathrm{image}(3\mathcal{O}(4)\oplus 8\mathcal{O}(5) \to 2\mathcal{O}(6)\oplus 6\mathcal{O}(7)).$$

and $\mathcal{G}^* \leftarrow 6\mathcal{O}(5)$ factors over $\mathcal{G}^* \leftarrow 8\mathcal{O}(5)\oplus 3\mathcal{O}(4)$. A general $\phi \in \mathrm{Hom}(6\mathcal{O}(5), \mathcal{G}^*)$ gives a point in $\mathbb{G}(6, 8)$ and the Hilbert scheme of desired curves would have dimension $36 + 12 = 48 = 4 \cdot 12 = 30 + 3 + 15$ as expected, c.f. [19].

Therefore the computation for obtaining a random space curve of genus 11 is done as follows:

```
i6 : randomGenus11Curve = (R) -> (
          correctCodimAndDegree:=false;
          while not correctCodimAndDegree do (
               Mt=coker random(R^{3:8},R^{6:7,2:6});
               M=coker (transpose (res Mt).dd_4);
               Gt:=transpose (res M).dd_3;
               I:=ideal syz (Gt*random(source Gt,R^{6:5}));
               correctCodimAndDegree=(codim I==2 and degree I==12););
          I);
```

In general for these problems there is rarely an a priori reason why such a construction for general choices will give a smooth curve. Kleiman's global generation condition [23] is much too strong a hypothesis for many interesting examples. But it is easy to check an example over a finite field with a computer:

```
i7 : isSmoothSpaceCurve = (I) -> (
         --I generates the ideal sheaf of a pure codim 2 scheme in P3
         singI:=I+minors(2,jacobian I);
         codim singI==4);
```

Hence by semi-continuity this is true over $\mathbb{Q}$ and the desired unirationality of $G(11,12,3)/\mathfrak{M}_{11}$ holds for all fields, except possibly for those whose ground field has characteristic is in some finite set.

A calculation of an example over the integers would bound the number of exceptional characteristics, which then can be ruled out case by case, or by considering sufficiently many integer examples.

As in case of nodal curves, to prove unirationality of $\mathfrak{M}_{11}$ by computer aided computations we have to show the injectivity of

$$H^0(L) \otimes H^0(K_C \otimes L^{-1}) \xrightarrow{\mu_L} H^0(K_C),$$

where L is the restriction of $\mathcal{O}_{\mathbb{P}^3}(1)$ to the curve $C \subset \mathbb{P}^3$. The following few lines do the job:

```
i8 : K=ZZ/101;

i9 : R=K[x_0..x_3];

i10 : C=randomGenus11Curve(R);

o10 : Ideal of R

i11 : isSmoothSpaceCurve(C)

o11 = true

i12 : Omega=prune Ext^2(coker gens C,R^{-4});

i13 : betti Omega

o13 = relations : total: 5 10
                     -1: 2  .
                      0: 3 10
```

We see that there are no linear relations among the two generators of $H^0_*(\Omega_C)$ in degree -1.

Betti Numbers for Genus $g = 12, 13, 14, 15$. The approach in these cases is similar to $g = 11$. We choose here $d = g$, so $\rho(g, g, 3) = g - 12 \geq 0$. Under the maximal rank assumption the corresponding space curve has a Hartshorne-Rao module whose Hilbert function takes values $(0, 0, g-9, 2g-19, 3g-34, 0, \ldots)$ in case $g = 12, 13$ and $(0, 0, g-9, 2g-19, 3g-34, 4g-55, 0, \ldots)$ in case $g = 14, 15$. Expected syzygies of M have Betti tables:

$g = 12:$

3	7	-	-	-
-	-	10	5	-
-	-	-	3	2

$g = 13:$

4	9	1	-	-
-	-	6	-	-
-	-	6	13	5

$g = 14:$

5	11	2	-	-
-	-	3	-	-
-	-	13	17	4
-	-	-	-	1

$g = 15:$

6	13	3	-	-
-	-	3	-	-
-	-	8	3	-
-	-	-	9	5

Comparing with the expected syzygies of C

$g = 12:$

1	-	-	-
-	-	-	-
-	-	-	-
-	-	-	-
-	7	5	-
-	-	3	2

$g = 13:$

1	-	-	-
-	-	-	-
-	-	-	-
-	-	-	-
-	3	-	-
-	6	13	5

$g = 14:$

1	-	-	-
-	-	-	-
-	-	-	-
-	-	-	-
-	-	-	-
-	13	17	4
-	-	-	1

$g = 15:$

1	-	-	-
-	-	-	-
-	-	-	-
-	-	-	-
-	-	-	-
-	8	3	-
-	-	9	5

we see that given M the choice of a curve corresponds to a point in $\mathbb{G}(7, 10)$ or $\mathbb{G}(3, 6)$ for $g = 12, 13$ respectively, while for $g = 14, 15$ everything is determined by the Hartshorne-Rao module. For $g = 12$ Kleiman's result guarantees smoothness for general choices, in contrast to the more difficult cases $g = 14, 15$. So the construction of M is the crucial step.

Construction of Hartshorne-Rao Modules. In case $g = 12$ the construction of M is straightforward. It is presented by a sufficiently general matrix of linear forms:

$$0 \leftarrow M \leftarrow 3S(-2) \leftarrow 7S(-3).$$

The procedure for obtaining a random genus 12 curve is:

```
i14 : randomGenus12Curve = (R) -> (
           correctCodimAndDegree:=false;
           while not correctCodimAndDegree do (
                M:=coker random(R^{3:-2},R^{7:-3});
                Gt:=transpose (res M).dd_3;
                I:=ideal syz (Gt*random(source Gt,R^{7:5}));
                correctCodimAndDegree=(codim I==2 and degree I==12););
           I);
```

In case $g = 13$ we have to make sure that M has a second linear syzygy. Consider the end of the Koszul complex:

$$6R(-2) \xleftarrow{\kappa} 4R(-3) \leftarrow R(-1) \leftarrow 0.$$

Any product of a general map $4R(-2) \xleftarrow{\alpha} 6R(-2)$ with the Koszul matrix κ yields $4R(-2) \leftarrow 4R(-3)$ with a linear syzygy, and concatenated with a general map $4R(-2) \xleftarrow{\beta} 5R(-3)$ gives a presentation matrix of a module M of desired type:

$$0 \leftarrow M \leftarrow 4R(-2) \leftarrow 4R(-3) \oplus 5R(-3).$$

The procedure for obtaining a random genus 13 curve is:

```
i15 : randomGenus13Curve = (R) -> (
           kappa:=koszul(3,vars R);
           correctCodimAndDegree:=false;
           while not correctCodimAndDegree do (
                test:=false;while test==false do (
                     alpha:=random(R^{4:-2},R^{6:-2});
                     beta:=random(R^{4:-2},R^{5:-3});
                     M:=coker(alpha*kappa|beta);
                     test=(codim M==4 and degree M==16););
                Gt:=transpose (res M).dd_3;
                --up to change of basis we can reduce phi to this form
                phi:=random(R^6,R^3)++id_(R^6);
                I:=ideal syz(Gt_{1..12}*phi);
                correctCodimAndDegree=(codim I==2 and degree I==13););
           I);
```

The case of genus $g = 14$ is about a magnitude more difficult. To start with, we can achieve two second linear syzygies by the same method as in the case $g = 13$. A general matrix $5R(-2) \xleftarrow{\alpha} 12R(-2)$ composed with $12R(-2) \xleftarrow{\kappa\oplus\kappa} 8R(-3)$ yields the first component of

$$5R(-2) \leftarrow (8+3)R(-3).$$

For a general choice of the second component $5R(-2) \xleftarrow{\beta} 3R(-3)$ the cokernel will be a module with Hilbert function $(0, 0, 5, 9, 8, 0, 0, \ldots)$ and syzygies

5	11	2	-	-
-	-	2	-	-
-	-	17	23	8
-	-	-	-	-

What we want is to find α and β so that $\dim M_5 = 1$ and $\dim \mathrm{Tor}_2^R(M, \mathbb{F})_5 = 3$. Taking into account that we ensured $\dim \mathrm{Tor}_2^R(M, \mathbb{F})_4 = 2$ this amounts to asking that the 100×102 matrix $m(\alpha, \beta)$ obtained from

$$[0 \leftarrow 5R(-2)_5 \leftarrow 11R(-3)_5 \leftarrow 2R(-4)_5 \leftarrow 0] \cong [0 \leftarrow 100\mathbb{F} \xleftarrow{m(\alpha,\beta)} 102\mathbb{F} \leftarrow 0]$$

drops rank by 1. We do not know a systematic approach to produce such $m(\alpha,\beta)$'s. However, we can find such matrices in a probabilistic way. In the space of the matrices $m(\alpha,\beta)$, those which drop rank by 1 have expected codimension 3. Hence over a finite field $\mathbb{F} = \mathbb{F}_q$ we expect to find the desired modules M with a probability of $1/q^3$. The code to detect bad modules is rather fast.

```
i16 : testModulesForGenus14Curves = (N,p) ->(
          x := local x;
          R := ZZ/p[x_0..x_3];
          i:=0;j:=0;
          kappa:=koszul(3,vars R);
          kappakappa:=kappa++kappa;
          utime:=timing while (i<N) do (
               test:=false;
               alpha:=random(R^{5:-2},R^{12:-2});
               beta:=random(R^{5:-2},R^{3:-3});
               M:=coker (alpha*kappakappa|beta);
               fM:=res (M,DegreeLimit =>3);
               if (tally degrees fM_2)_{5}==3 then (
                    --further checks to pick up the right module
                    test=(tally degrees fM_2)_{4}==2 and
                    codim M==4 and degree M==23;);
               i=i+1;if test==true then (j=j+1;););
          timeForNModules:=utime#0; numberOfGoodModules:=j;
          {timeForNModules,numberOfGoodModules});

i17 : testModulesForGenus14Curves(1000,5)

o17 = {41.02, 10}

o17 : List
```

For timing tests we used a Pentium2 400Mhz with 128Mb of memory running GNU Linux. On such a machine examples can be tested at a rate of 0.04 seconds per example. Hence an approximate estimation of the CPU-time required to find a good example is $q^3 \cdot 0.04$ seconds. Comparing this with the time to verify smoothness, which is about 12 seconds for an example of this degree, we see that up to $|\mathbb{F}_q| = q \leq 13$ we can expect to obtain examples within few minutes. Actually the computations for $q = 2$ and $q = 3$ take longer than for $q = 5$ on average, because examples of "good modules" tend to give singular curves more often. Here is a table of statistics which summarizes the situation.

q	2	3	5	7	11	13
smooth curves	100	100	100	100	100	100
1-nodal curves	75	53	31	16	10	8
reduced more singular	1012	142	24	11	2	0
non reduced curves	295	7	0	0	0	0
total number of curves	1482	302	155	127	112	108
percentage of smooth curves	6.7%	33%	65%	79%	89%	93%
approx. time (in seconds)	7400	3100	2700	3400	6500	9500

The procedure for obtaining a random genus 14 curve is

```
i18 : randomGenus14Curve = (R) -> (
          kappa:=koszul(3,vars R);
          kappakappa:=kappa++kappa;
          correctCodimAndDegree:=false;
          count:=0;while not correctCodimAndDegree do (
               test:=false;
               t:=timing while test==false do (
                    alpha=random(R^{5:-2},R^{12:-2});
                    beta=random(R^{5:-2},R^{3:-3});
                    M:=coker (alpha*kappakappa|beta);
                    fM:=res (M,DegreeLimit =>3);
                    if (tally degrees fM_2)_{5}==3 then (
                         --further checks to pick up the right module
                         test=(tally degrees fM_2)_{4}==2 and
                         codim M==4 and degree M==23;);
                    count=count+1;);
               Gt:=transpose (res M).dd_3;
               I:=ideal syz (Gt_{5..17});
               correctCodimAndDegree=(codim I==2 and degree I==14););
          <<"     -- "<<t#0<<" seconds used for ";
          <<count<<" modules"<<endl;
          I);
```

For $g = 15$ we do not know a method along these lines that would give examples over small fields.

Counting Parameters. For genus $g = 12$ clearly the module M depends on $\dim \mathbb{G}(7, 3 \cdot h^0\mathcal{O}(1)) - \dim SL(3) = 7 \cdot 5 - 8 = 36$ parameters, and the family of curves has dimension $36 + \dim \mathbb{G}(7, 10) = 48 = 4 \cdot 12 = 33 + 0 + 15$, as expected.

For genus $g = 13$ and 14 the parameter count is more difficult. Let us make a careful parameter count for genus $g = 14$; the case $g = 13$ is similar and easier. The choice of α corresponds to a point in $\mathbb{G}(5, 12)$. Then β corresponds to a point $\mathbb{G}(3, B_\alpha)$ where $B_\alpha = U \otimes R_1/{<}\alpha{>}$ where U denotes the universal subbundle on $\mathbb{G}(5, 12)$ and ${<}\alpha{>}$ the subspace generated by the 8 columns of $\alpha \circ (\kappa \oplus \kappa)$. So $\dim B_\alpha = 20 - 8 = 12$ and $\mathbb{G}(3, B_\alpha) \to \mathbb{G}(5, 12)$ is a Grassmannian bundle with fiber dimension 27 and total dimension 62. In this space the scheme of good modules has codimension 3, so we get a 59 dimensional family. This is larger than the expected dimension $56 = 4 \cdot 14 = 39 + 2 + 15$ of the Hilbert scheme, c.f. [19]. Indeed the construction gives a curve together with a basis of $\mathrm{Tor}_2^R(M, \mathbb{F})_4$. Subtracting the dimension of the group of the projective coordinate changes we arrive at the desired dimension $59 - 3 = 56$.

The unirationality of $\mathfrak{M}_{12}$ and $\mathfrak{M}_{13}$ can be proved by computer as in case $\mathfrak{M}_{11}$, while in case $g = 14$ we don't know the unirationality of the parameter space of the modules M with $\dim M_5 = 1$ and $\dim \mathrm{Tor}_2^R(M, \mathbb{F})_5 = 3$.

2 Comparing Green's Conjecture for Curves and Points

2.1 Syzygies of Canonical Curves

One of the most outstanding conjectures about free resolutions is Green's prediction for the syzygies of canonical curves.

A canonical curve $C \subset \mathbb{P}^{g-1}$, i.e., a linearly normal curve with $\mathcal{O}_C(1) \equiv \omega_C$, the canonical line bundle, is projectively normal by a result of Max Noether, and hence has a Gorenstein homogeneous coordinate ring and is 3-regular.

Therefore the Betti numbers of the free resolution of a canonical curve are symmetric, that is, $\beta_{j,j+1} = \beta_{g-2-j,g-j}$, and essentially only two rows of Betti numbers occur. The situation is summarized in the following table.

1	-	-	-	-	-	-	-	-	-	-	-
-	$\beta_{1,2}$	$\beta_{2,3}$	$\cdots$	$\beta_{p,p+1}$	$\cdots$	$\bullet$ $\cdots$	$\beta_{p,p+2}$	-	-	-	-
-	-	-	-	$\beta_{p,p+2}$	$\cdots$	$\cdots$	$\beta_{p,p+1}$	$\cdots$	$\beta_{2,3}$	$\beta_{1,2}$	-
-	-	-	-	-	-	-	-	-	-	-	1

The first p such that $\beta_{p,p+2} \neq 0$ is conjecturally precisely the Clifford index of the curve.

Conjecture 2.1 (Green [16]). Let C be a smooth canonical curve over $\mathbb{C}$. Then $\beta_{p,p+2} \neq 0$ if and only if $\exists L \in \operatorname{Pic}^d(C)$ with $h^0(C,L), h^1(C,L) \geq 2$ and $\operatorname{cliff}(L) := d - 2(h^0(C,L)) - 1) \leq p$. In particular, $\beta_{j,j+2} = 0$ for $j \leq \lfloor \frac{g-3}{2} \rfloor$ for a general curve of genus g.

The "if" part is proved by Green and Lazarsfeld in [18] and holds for arbitrary ground fields. For some partial results see [35,31,30,4,5,22,25]. The conjecture is known to be false for some (algebraically closed) fields of finite characteristic, e.g., genus $g = 7$ and characteristic $\operatorname{char} \mathbb{F} = 2$; see [29].

2.2 Coble Self-Dual Sets of Points

The free resolution of a hyperplane section of a Cohen-Macaulay ring has the same Betti numbers. Thus we may ask for a geometric interpretation of the syzygies of $2g-2$ points in $\mathbb{P}^{g-2}$ (hyperplane section of a canonical curve), or syzygies of a graded Artinian Gorenstein algebra with Hilbert function $(1, g-2, g-2, 1, 0, \ldots)$ (twice a hyperplane section). Any collection of $2g-2$ points obtained as a hyperplane section of a canonical curve is special in the sense that it imposes only $2g-3$ conditions on quadrics. An equivalent condition for points in linearly uniform position is that they are Coble (or Gale) self-dual; see [14]. Thus if we distribute the $2g-2$ points into two collections each of $g-1$ points, with, say, the first consisting of the coordinate points and the second corresponding to the rows of a $(g-1) \times (g-1)$ matrix

$A = (a_{ij})$, then A can be chosen to be an orthogonal matrix, i.e., $A^t A = 1$; see [14].

To see what the analogue of Green's Conjecture for the general curve means for orthogonal matrices we recall a result of [28].

Set $n = g - 2$. We identify the homogeneous coordinate ring of $\mathbb{P}^n$ with the ring $S = \mathbb{F}[\partial_0, \ldots, \partial_n]$ of differential operators with constant coefficients, $\partial_i = \frac{\partial}{\partial x_i}$. The ring S acts on $\mathbb{F}[x_0, \ldots, x_n]$ by differentiation. The annihilator of $q = x_0^2 + \ldots + x_n^2$ is a homogeneous ideal $J \subset S$ such that S/J is a graded Artinian Gorenstein ring with Hilbert function $(1, n+1, 1)$ and socle induced by q, see [27], [12, Section 21.2 and related exercise 21.7]. The syzygy numbers of S/J are

1	-	-	-	-	-	-
-	$\frac{n}{n+2}\binom{n+3}{2}$	$\cdots$	$\frac{p(n+1-p)}{n+2}\binom{n+3}{p+1}$	$\cdots$	$\frac{n}{n+2}\binom{n+3}{n+1}$	-
-	-	-	-	-	-	1

A collection $H_0, \ldots, H_n$ of hyperplanes in $\mathbb{P}^n$ is said to form a polar simplex to q if and only if the collection $\Gamma = \{p_0, \ldots, p_n\} \subset \check{\mathbb{P}}^n$ of the corresponding points in the dual space has its homogeneous ideal $I_\Gamma \subset S$ contained in J.

In particular the set Λ consisting of the coordinate points correspond to a polar simplex, because $\partial_i \partial_j$ annihilates q for $i \neq j$.

For any polar collection of points Γ the free resolution $\mathbf{S}_\Gamma$ is a subcomplex of the resolution $\mathbf{S}_{S/J}$. Green's conjecture for the generic curve of genus $g = n + 2$ would imply:

Conjecture 2.2. For a general Γ and the given Λ the corresponding Tor-groups

$$\mathrm{Tor}_k^S(S/I_\Gamma, \mathbb{F})_{k+1} \cap \mathrm{Tor}_k^S(S/I_\Lambda, \mathbb{F})_{k+1} \subset \mathrm{Tor}_k^S(S/J, \mathbb{F})_{k+1}$$

intersect transversally.

Proof. A zero-dimensional non-degenerate scheme $\Gamma \subset \mathbb{P}^n$ of degree n+1 has syzygies

1	-	-	-	-	-
-	$\binom{n+1}{2}$	$\cdots$	$k\binom{n+1}{k+1}$	$\cdots$	n

Since both Tor groups are contained in $\mathrm{Tor}_k^S(S/J, \mathbb{F})_{k+1}$, the claim is equivalent to saying that for a general polar simplex Γ the expected dimension of their intersection is $\dim \mathrm{Tor}_k^S(S/_\Gamma, \mathbb{F})_{k+1} + \dim \mathrm{Tor}_k^S(S/_\Gamma, \mathbb{F})_{k+1} - \dim \mathrm{Tor}_k^S(S/J, \mathbb{F})_{k+1}$, which is

$$2k\binom{g-1}{k+1} - \frac{k(g-1-k)}{g}\binom{g+1}{k+1}.$$

On the other hand, $I_{\Gamma\cup\Lambda} = I_\Gamma \cap I_\Lambda$, hence

$$\operatorname{Tor}_k^S(S/I_\Gamma, \mathbb{F})_{k+1} \cap \operatorname{Tor}_k^S(S/I_\Lambda, \mathbb{F})_{k+1} = \operatorname{Tor}_k^S(S/I_{\Gamma\cup\Lambda}, \mathbb{F})_{k+1},$$

and Green's conjecture would imply

$$\dim \operatorname{Tor}_k^S(S/I_{\Gamma\cup\Lambda}, \mathbb{F})_{k+1} = \beta_{k,k+1}(\Gamma \cup \Lambda) = k\binom{g-2}{k+1} - (g-1-k)\binom{g-2}{k-2}.$$

Now a calculation shows that the two dimensions above are equal.

The family of all polar simplices V is dominated by the family defined by the ideal of 2×2 minors of the matrix

$$\begin{pmatrix} \partial_0 & \dots & \partial_i & \dots & \partial_n \\ \sum_j b_{0j}\partial_j & \dots & \sum_j b_{ij}\partial_j & \dots & \sum_j b_{nj}\partial_j \end{pmatrix}$$

depending on a symmetric matrix $B = (b_{ij})$, i.e., $b_{ij} = b_{ji}$ as parameters. For B a general diagonal matrix we get Λ together with a specific element in $\operatorname{Tor}_n^S(S/I_\Lambda, \mathbb{F})_{n+1}$.

2.3 Comparison and Probes

One of the peculiar consequences of Green's conjecture for odd genus $g = 2k+1$ is that, if $\beta_{k,k+1} = \beta_{k-1,k+1} \neq 0$, then the curve C lies in the closure of the locus of $k+1$-gonal curve. Any $k+1$-gonal curve lies on a rational normal scroll X of codimension k that satisfies $\beta_{k,k+1}(X) = k$. Hence

$$\beta_{k,k+1}(C) \neq 0 \Rightarrow \beta_{k,k+1}(C) \geq k$$

We may ask whether a result like this is true for the union of two polar simplices $\Lambda \cup \Gamma \subset \mathbb{P}^{2k-1}$. Define

$$\tilde{D} = \{\Gamma \in V | \Gamma \cup \Lambda \text{ is syzygy special}\}$$

where, as above, V denotes the variety of polar simplices and Λ denotes the coordinate simplex. If $\tilde{D}$ is a proper subvariety, then it is a divisor, because $\beta_{k,k+1}(\Gamma \cup \Lambda) = \beta_{k-1,k+1}(\Gamma \cup \Lambda)$.

Conjecture 2.3. The subscheme $\tilde{D} \subset V$ is an irreducible divisor, for $g = n+2 = 2k+1 \in \{7, 9, 11\}$. The value of $\beta_{k,k+1}$ on a general point of D is $3, 2, 1$ respectively.

We can prove this for $g = 7$ by computer algebra. For $g = 9$ and $g = 11$ a proof is computationally out of reach with our methods, but we can get some evidence from examples over finite fields.

Evidence. Since $\tilde{D}$ is a divisor, we expect that if we pick symmetric matrices B over $\mathbb{F}_q$ at random, we will hit points on every component of $\tilde{D}$ at a probability of $1/q$. For a general point on $\tilde{D}$ the corresponding Coble self-dual set of points will have the generic number of extra syzygies of that component. Points with even more syzygies will occur in higher codimension, hence only with a probability of $1/q^2$. Some evidence for irreducibility can be obtained from the Weil formulas: for sufficiently large q we should see points on $\tilde{D}$ with probability $Cq^{-1}+O(q^{-\frac{3}{2}})$, where C is the number of components.

The following tables give for small fields $\mathbb{F}_q$ the number s_i of examples with i extra syzygies in a test of 1000 examples for $g = 9$ and 100 examples for $g = 11$. The number $s_{\text{tot}} = \sum_{i>0} s_i$ is the total number of examples with extra syzygies.
Genus $g = 9$:

q	$1000/q$	s_{tot}	s_1	s_2	s_3	s_4
2	500	925	0	130	0	63
3	333	782	0	273	0	33
4	250	521	0	279	0	99
5	200	350	0	217	0	74
7	143	197	0	144	0	36
8	125	199	0	147	0	43
9	111	218	0	98	0	0
11	91	118	0	102	0	15
13	77	90	0	79	0	10
16	62	72	0	68	0	4
17	59	76	0	69	0	6

Genus $g = 11$:

q	$100/q$	s_{tot}	s_1	s_2	s_3	s_4
7	14.3	16	14	0	0	0
17	5.9	7	7	0	0	0

In view of these numbers, it is more likely that the set $\tilde{D}$ of syzygy special Coble points is irreducible than that it is reducible. For a more precise statement we refer to [6].

A test of Green's Conjecture for Curves. In view of 2.3 it seems plausible that for a general curve of odd genus $g \geq 11$ with $\beta_{k,k+1}(C) \neq 0$ the value might be $\beta_{k,k+1} = 1$ contradicting Green's conjecture. It is clear that the syzygy exceptional locus has codimension 1 in $\mathfrak{M}_g$ for odd genus, if it is proper, i.e., if Green's conjecture holds for the general curve of that genus. So picking points at random we might be able to find such curve over a finite field $\mathbb{F}_q$ with probability $1/q$, roughly.

Writing code that does this is straightforward. One makes a loop that picks up randomly a curve, computes its canonical image, and resolves its ideal, counting the possible values $\beta_{k,k+1}$ until a certain amount of special curves is reached. The result for 10 special curves in $\mathbb{F}_7$ is as predicted:

g	seconds	total curves	special curves	possible values of $\beta_{k,k+1}$ ≤ 2	3	4	5	6	≥ 7
7	148	75	10	0	10	0	0	0	0
9	253	58	10	0	0	9	0	0	1
11	25640	60	10	0	0	0	9	0	1

(The test for genus 9 and 11 used about 70 and 120 megabytes of memory, respectively.)

So Green's conjecture passed the test for $g = 9, 11$. Shortly after the first author tried this test for the first time, a paper of Hirschowitz and Ramanan appeared proving this in general:

Theorem 2.4 ([22]). *If the general curve of odd genus $g = 2k+1$ satisfies Green's conjecture then the syzygy special curves lie on the divisor $D = \{C \in \mathfrak{M}_g | W^1_{k+1}(C) \neq \emptyset\}$*

The theorem gives strong evidence for the full Green's conjecture in view of our study of Coble self-dual sets of points.

Our findings suggest that the variety of points arising as hyperplane sections of smooth canonical curves has the strange property that it intersects the divisor of syzygy special sets of points $\tilde{D}$ only in its singular locus.

The conjecture for general curves is known to us up to $g \leq 17$, which is as far as a computer allows us to do a ribbon example; see [4].

3 Pfaffian Calabi-Yau Threefolds in $\mathbb{P}^6$

Calabi-Yau 3-folds caught the attention of physicists because they can serve as the compact factor of the Kaluza-Klein model of spacetime in superstring theory. One of the remarkable things that grows out of the work in physics is the discovery of mirror symmetry, which associates to a family of Calabi-Yau 3-folds (M_λ), another family (W_μ) whose Hodge diamond is the mirror of the Hodge diamond of the original family.

Although there is an enormous amount of evidence at present, the existence of a mirror is still a hypothesis for general Calabi-Yau 3-folds. The thousands of cases where this was established all are close to toric geometry, where through the work of Batyrev and others [3,9] rigorous mirror constructions were given and parts of their conjectured properties proved.

From a commutative algebra point of view the examples studied so far are rather trivial, because nearly all are hypersurfaces or complete intersections on toric varieties, or zero loci of sections in homogeneous bundles on homogeneous spaces.

Of course only a few families of Calabi-Yau 3-folds should be of this kind. Perhaps the easiest examples beyond the toric/homogeneous range are Calabi-Yau 3-folds in $\mathbb{P}^6$. Here examples can be obtained by the Pfaffian construction of Buchsbaum-Eisenbud [7] with vector bundles; see section 3.1 below. Indeed a recent theorem of Walter [36] says that any smooth Calabi-Yau in $\mathbb{P}^6$ can be obtained in this way. In this section we report on our construction of such examples.

As is quite usual in this kind of problem, there is a range where the construction is still quite easy, e.g., for surfaces in $\mathbb{P}^4$ the work in [10,26] shows that the construction of nearly all the 50 known families of smooth non-general type surfaces is straight forward and their Hilbert scheme component unirational. Only in very few known examples is the construction more difficult and the unirationality of the Hilbert scheme component an open problem.

The second author did the first "non-trivial" case of a construction of Calabi-Yau 3-folds in $\mathbb{P}^6$. Although in the end the families turned out to be unirational, the approach utilized small finite field constructions as a research tool.

3.1 The Pfaffian Complex

Let $\mathcal{F}$ be a vector bundle of odd rank $\operatorname{rk}\mathcal{F} = 2r+1$ on a projective manifold M, and let $\mathcal{L}$ be a line bundle. Let $\varphi \in H^0(M, \Lambda^2\mathcal{F} \otimes \mathcal{L})$ be a section. We can think of φ as a skew-symmetric twisted homomorphism

$$\mathcal{F}^* \xrightarrow{\varphi} \mathcal{F} \otimes \mathcal{L}.$$

The r^{th} divided power of φ is a section $\varphi^{(r)} = \frac{1}{r!}(\varphi \wedge \cdots \wedge \varphi) \in H^0(M, \Lambda^{2r}\mathcal{F} \otimes \mathcal{L}^r)$. Wedge product with $\varphi^{(r)}$ defines a morphism

$$\mathcal{F} \otimes \mathcal{L} \xrightarrow{\psi} \Lambda^{2r+1}\mathcal{F} \otimes \mathcal{L}^{r+1} = \det(\mathcal{F}) \otimes \mathcal{L}^{r+1}.$$

The twisted image $\mathcal{I} = \operatorname{image}(\psi) \otimes \det(\mathcal{F}^*) \otimes \mathcal{L}^{-r-1} \subset \mathcal{O}_M$ is called the *Pfaffian ideal* of φ, because working locally with frames, it is given by the ideal generated by the $2r \times 2r$ principle Pfaffians of the matrix describing φ. Let $\mathcal{D}$ denote the determinant line bundle $\det(\mathcal{F}^*)$.

Theorem 3.1 (Buchsbaum-Eisenbud [7]). *With this notation*

$$0 \to \mathcal{D}^2 \otimes \mathcal{L}^{-2r-1} \xrightarrow{\psi^t} \mathcal{D} \otimes \mathcal{L}^{-r-1} \otimes \mathcal{F}^* \xrightarrow{\varphi} \mathcal{F} \otimes \mathcal{D} \otimes \mathcal{L}^{-r} \xrightarrow{\psi} \mathcal{O}_M$$

is a complex. $X = V(\mathcal{I}) \subset M$ *has codimension* ≤ 3 *at every point, and in case equality holds (everywhere along* X*) then this complex is exact and resolves the structure sheaf* $\mathcal{O}_X = \mathcal{O}_M/\mathcal{I}$ *of the locally Gorenstein subscheme* X*.*

We will apply this to construct Calabi-Yau 3-folds in $\mathbb{P}^6$. In that case we want X to be smooth and $\det(\mathcal{F})^{-2} \otimes \mathcal{L}^{-2r-1} \cong \omega_{\mathbb{P}} \cong \mathcal{O}(-7)$, so we may conclude that $\omega_X \cong \mathcal{O}_X$. A result of Walter [36] for $\mathbb{P}^n$ guarantees the existence of a Pfaffian presentation in $\mathbb{P}^6$ for every subcanonical embedded 3-fold. Moreover Walter's choice of $\mathcal{F}\otimes\mathcal{D}\otimes\mathcal{L}^{-r}$ for Calabi-Yau 3-folds $X \subset \mathbb{P}^6$ is the sheafified first syzygy module $H^1_*(\mathcal{I}_X)$ plus possibly a direct sum of line bundles (indeed $H^2_*(\mathcal{I}_X) = 0$ because of the Kodaira vanishing theorem). Under the maximal rank assumption for

$$H^0(\mathbb{P}^6, \mathcal{O}(m)) \to H^0(X, \mathcal{O}_X(m))$$

the Hartshorne-Rao module is zero for $d = \deg X \in \{12, 13, 14\}$ and an arithmetically Cohen-Macaulay X is readily found. For $d \in \{15, 16, 17, 18\}$ the Hartshorne-Rao modules M have Hilbert functions with values $(0, 0, 1, 0, \ldots)$, $(0, 0, 2, 1, 0, \ldots)$, $(0, 0, 3, 5, 0, \ldots)$ and $(0, 0, 4, 9, 0, \ldots)$ respectively.

We do not discuss the cases $d \leq 16$ further. The construction in those cases is obvious; see [34].

3.2 Analysis of the Hartshorne-Rao Module for Degree 17

Denote with $\mathcal{F}_1$ the sheaf $\mathcal{F}\otimes\mathcal{D}\otimes\mathcal{L}^{-r}$. We try to construct $\mathcal{F}_1$ as the sheafified first syzygy module of M. The construction of a module with the desired Hilbert function is straightforward. The cokernel of $3S(-2) \xleftarrow{b} 16S(-3)$ for a general matrix of linear forms has this property. However, for a general b and $\mathcal{F}_1 = \ker(16\mathcal{O}(-3) \xrightarrow{b} 3\mathcal{O}(-2))$ the space of skew-symmetric maps $\mathrm{Hom}_{\mathrm{skew}}(\mathcal{F}_1^*(-7), \mathcal{F}_1)$ is zero: M has syzygies

3	16	28	-	-	-	-	
-	-	-	70	112	84	32	5

Any map $\varphi\colon \mathcal{F}_1^*(-7) \to \mathcal{F}_1$ induces a map on the free resolutions:

$$\begin{array}{ccccccccc}
0 & \longleftarrow & \mathcal{F}_1 & \longleftarrow & 28\mathcal{O}(-4) & \longleftarrow & 70\mathcal{O}(-6) & \longleftarrow & 112\mathcal{O}(-7) \\
 & & \uparrow{\scriptstyle\varphi} & & \uparrow{\scriptstyle\varphi_0} & & \uparrow{\scriptstyle\varphi_1} & & \\
0 & \longleftarrow & \mathcal{F}_1^*(-7) & \longleftarrow & 16\mathcal{O}(-4) & \longleftarrow & 3\mathcal{O}(-5) & \longleftarrow & 0
\end{array}$$

Since $\varphi_1 = 0$ for degree reasons, $\varphi = 0$ as well, and $\mathrm{Hom}(\mathcal{F}_1^*(-7), \mathcal{F}_1) = 0$ for a general module M.

What we need are special modules M that have extra syzygies

3	16	28	k	-	-	-	
-	-	k	70	112	84	32	5

with k at least 3.

In a neighborhood of $o \in \operatorname{Spec} B$, where denotes B the base space of a semi-universal deformation of M, the resolution above would lift to a complex over $B[x_0, \ldots, x_6]$ and in the lifted complex there is a $k \times k$ matrix Δ with entries in the maximal ideal $o \subset B$. By the principal ideal theorem we see that Betti numbers stay constant in a subvariety of codimension at most k^2. To check for second linear syzygies on a randomly chosen M is a computationally rather easy task. The following procedure tests the computer speed of this task.

```
i19 : testModulesForDeg17CY = (N,k,p) -> (
           x:=symbol x;R:=(ZZ/p)[x_0..x_6];
           numberOfGoodModules:=0;i:=0;
           usedTime:=timing while (i<N) do (
                b:=random(R^3,R^{16:-1});
                --we put SyzygyLimit=>60 because we expect
                --k<16 syzygies, so 16+28+k<=60
                fb:=res(coker b,
                     DegreeLimit =>0,SyzygyLimit=>60,LengthLimit =>3);
                if rank fb_3>=k and (dim coker b)==0 then (
                     fb=res(coker b, DegreeLimit =>0,LengthLimit =>4);
                     if rank fb_4==0
                     then numberOfGoodModules=numberOfGoodModules+1;);
                i=i+1;);
           collectGarbage();
           timeForNModules:=usedTime#0;
           {timeForNModules,numberOfGoodModules});
```

Running this procedure we see that it takes not more than 0.64 seconds per example. Hence we can hope to find examples with $k = 3$ within a reasonable time for a very small field, say $\mathbb{F}_3$.

The first surprise is that examples with k extra syzygies are found much more often, as can be seen by looking at the second value output by the function `testModulesForDeg17CY()`.

This is not only a "statistical" remark, in the sense that the result is confirmed by computing the semi-universal deformations of these modules. Indeed define $\mathbb{M}_k = \{M \mid \operatorname{Tor}_3^S(M, \mathbb{F})_5 \geq k\}$ and consider a module $M \in \mathbb{M}_k$: "generically" we obtain $\operatorname{codim}(\mathbb{M}_k)_M = k$ instead of k^2 (and in fact one can diagonalize the matrix Δ over the algebraic closure $\bar{\mathbb{F}}$).

The procedure is straightforward but a bit long. First we pick up an example with k-extra syzygies.

```
i20 : randomModuleForDeg17CY = (k,R) -> (
           isGoodModule:=false;i:=0;
           while not isGoodModule do (
                b:=random(R^3,R^{16:-1});
                --we put SyzygyLimit=>60 because we expect
                --k<16 syzygies, so 16+28+k<=60
                fb:=res(coker b,
                     DegreeLimit =>0,SyzygyLimit=>60,LengthLimit =>3);
                if rank fb_3>=k and (dim coker b)==0 then (
                     fb=res(coker b, DegreeLimit =>0,LengthLimit =>4);
                     if rank fb_4==0 then isGoodModule=true;);
                i=i+1;);
           <<"     -- Trial n. " << i <<", k="<< rank fb_3 <<endl;
```

```
          b);
```

Notice that the previous function returns a presentation matrix b of M, and not M.

Next we compute the tangent codimension of $\mathbb{M}_k$ in the given example $M = \operatorname{Coker} b$ by computing the codimension of the space of the infinitesimal deformations of M that still give an element in $\mathbb{M}_k$. Denote with b_i the maps in the linear strand of a minimal free resolution of M, and with b_2' the quadratic part in the second map of this resolution. Over $B = \mathbb{F}[\epsilon]/\epsilon^2$ let $b_1 + \epsilon f_1$ be an infinitesimal deformation of b_1. Then f_1 lifts to a linear map $f_2\colon 28S(-4) \to 16S(-3)$ determined by $(b_1+\epsilon f_1)\circ(b_2+\epsilon f_2) = 0$, and f_2 to a map $f_3 \oplus \Delta\colon kS(-5) \to 28S(-4) \oplus kS(-5)$ determined by $(b_2+b_2')\circ\epsilon(f_3\oplus\Delta) = 0$. Therefore we can determine Δ as:

```
i21 : getDeltaForDeg17CY = (b,f1) -> (
          fb:=res(coker b, LengthLimit =>3);
          k:=numgens target fb.dd_3-28; --# of linear syzygies
          b1:=fb.dd_1;b2:=fb.dd_2_{0..27};b2':=fb.dd_2_{28..28+k-1};
          b3:=fb.dd_3_{0..k-1}^{0..27};
          --the equation for f2 is b1*f2+f1*b2=0,
          --so f2 is a lift of (-f1*b2) through b1
          f2:=-(f1*b2)//b1;
          --the equation for A=(f3||Delta) is -f2*b3 = (b2|b2') * A
          A:=(-f2*b3)//(b2l|b2');
          Delta:=A^{28..28+k-1});
```

Now we just parametrize all possible maps $f_1\colon 16S(-3) \to 3S(-2)$, compute their respective maps Δ, and find the codimension of the condition that Δ is the zero map:

```
i22 : codimInfDefModuleForDeg17CY = (b) -> (
          --we create a parameter ring for the matrices f1's
          R:=ring b;K:=coefficientRing R;
          u:=symbol u;U:=K[u_0..u_(3*16*7-1)];
          i:=0;while i<3 do (
               <<endl<< " " << i+1 <<":" <<endl;
               j:=0;while j<16 do(
                    << "    " << j+1 <<". "<<endl;
                    k:=0;while k<7 do (
                       l=16*7*i+7*j+k; --index parametrizing the f1's
                       f1:=matrix(R,apply(3,m->apply(16,n->
                            if m==i and n==j then x_k else 0)));
                       Delta:=substitute(getDeltaForDeg17CY(b,f1),U);
                       if l==0 then (equations=u_l*Delta;) else (
                            equations=equations+u_l*Delta;);
                       k=k+1;);
                    collectGarbage(); --frees up memory in the stack
                    j=j+1;);
               i=i+1;);
          codim ideal equations);
```

The second surprise is that for $\mathcal{F}_1 = syz_1(M)$ we find

$$\dim \operatorname{Hom}_{\text{skew}}(\mathcal{F}_1^*(-7), \mathcal{F}_1) = k = \dim \operatorname{Tor}_3^S(M, \mathbb{F})_5.$$

$\operatorname{Hom}_{\text{skew}}(\mathcal{F}_1^*(-7), \mathcal{F}_1)$ is the vector space of skew-symmetric linear matrices φ such that $b \circ \varphi = 0$. The following procedure gives a matrix of size $\binom{16}{2} \times$

$\dim \mathrm{Hom}_{\mathrm{skew}}(\mathcal{F}_1^*(-7), \mathcal{F}_1)$ whose i-th column gives the entries of a 16×16 skew-symmetric matrix inducing the i-th basis element of the vector space $\mathrm{Hom}_{\mathrm{skew}}(\mathcal{F}_1^*(-7), \mathcal{F}_1)$.

```
i23 : skewSymMorphismsForDeg17CY = (b) -> (
          --we create a parameter ring for the morphisms:
          K:=coefficientRing ring b;
          u:=symbol u;U:=K[u_0..u_(binomial(16,2)-1)];
          --now we compute the equations for the u_i's:
          UU:=U**ring b;
          equationsInUU:=flatten (substitute(b,UU)*
               substitute(genericSkewMatrix(U,u_0,16),UU));
          uu:=substitute(vars U,UU);
          equations:=substitute(
               diff(uu,transpose equationsInUU),ring b);
          syz(equations,DegreeLimit =>0));
```

A morphism parametrized by a column `skewSymMorphism` is then recovered by the following code.

```
i24 : getMorphismForDeg17CY = (SkewSymMorphism) -> (
          u:=symbol u;U:=K[u_0..u_(binomial(16,2)-1)];
          f=map(ring SkewSymMorphism,U,transpose SkewSymMorphism);
          f genericSkewMatrix(U,u_0,16));
```

Rank 1 Linear Syzygies of M. To understand this phenomenon we consider the multiplication tensor of M:

$$\mu \colon M_2 \otimes V \to M_3$$

where $V = H^0(\mathbb{P}^6, \mathcal{O}(1))$.

Definition 3.2. A decomposable element of $M_2 \otimes V$ in the kernel of μ is called a rank 1 linear syzygy of M. The (projective) space of rank 1 syzygies is

$$Y = (\mathbb{P}^2 \times \mathbb{P}^6) \cap \mathbb{P}^{15} \subset \mathbb{P}^{20}$$

where $\mathbb{P}^2 = \mathbb{P}(M_2^*), \mathbb{P}^6 = \mathbb{P}(V^*)$ and $\mathbb{P}^{15} = \mathbb{P}(\ker(\mu)^*)$ inside the Segre space $\mathbb{P}((M_2 \otimes V)^*) \cong \mathbb{P}^{20}$.

Proposition 1.5 of [17] says that, for $\dim M_2 \leq j$, the existence of a j^{th} linear syzygy implies $\dim Y \geq j-1$. This is automatically satisfied for $j=3$ in our case: $\dim Y \geq 3$ with equality expected.

The projection $Y \to \mathbb{P}^2$ has linear fibers, and the general fiber is a $\mathbb{P}^1$. However, special fibers might have higher dimension. In terms of the presentation matrix b a special 2-dimensional fiber (defined over $\mathbb{F}$) corresponds to a block

$$b = \begin{pmatrix} 0 & 0 & 0 & * \dots \\ 0 & 0 & 0 & * \dots \\ l_1 & l_2 & l_3 & * \dots \end{pmatrix},$$

where l_1, l_2, l_3 are linear forms, in the 3×16 presentation matrix of M. Such a block gives a

```
1 3 3 1 - - - -
- - - - - - - -
```

subcomplex in the free resolution of M and an element $s \in H^0(\mathbb{P}^6, \Lambda^2\mathcal{F}_1 \otimes \mathcal{O}(7))$ since the syzygy matrix

$$\begin{pmatrix} 0 & -l_3 & l_2 \\ l_3 & 0 & -l_1 \\ -l_2 & l_1 & 0 \end{pmatrix}$$

is skew.

This answers the questions posed by both surprises: we want a module M with at least $k \geq 3$ special fibers and these satisfy $h^0(\mathbb{P}^6, \Lambda^2\mathcal{F}_1 \otimes \mathcal{O}(7)) \geq k$, if the k sections are linearly independent. The condition for k special fibers is of expected codimension k in the parameter space $\mathbb{G}(16, 3h^0(\mathbb{P}^6, \mathcal{O}(1))$ of the presentation matrices. In a given point M the actual codimension can be readily computed by a first order deformations and that $H^0(\mathbb{P}^6, \Lambda^2\mathcal{F}_1 \otimes \mathcal{O}(7))$ is k-dimensional, and spanned by the k sections corresponding to the k special fibers can be checked as well.

First we check that M has k distinct points in $\mathbb{P}(M_2^*)$ where the multiplication map drops rank. (Note that this condition is likely to fail over small fields. However, the check is computationally easy).

```
i25 : checkBasePtsForDeg17CY = b -> (
          --firstly the number of linear syzygies
          fb:=res(coker b, DegreeLimit=>0, LengthLimit =>4);
          k:=#select(degrees source fb.dd_3,i->i=={3});
          --then the check
          a=symbol a;A=K[a_0..a_2];
          mult:=(id_(A^7)**vars A)*substitute(
               syz transpose jacobian b,A);
          basePts=ideal mingens minors(5,mult);
          codim basePts==2 and degree basePts==k and distinctPoints(
               basePts));
```

Next we check that $H^0(\mathbb{P}^6, \Lambda^2\mathcal{F}_1 \otimes \mathcal{O}(7))$ is k-dimensional, by looking at the numbers of columns of `skewSymMorphismsForDeg17CY(b)`. Finally we do the computationally hard part of the check, which is to verify that the k special sections corresponding to the k special fibers of $Y \to \mathbb{P}^2$ span $H^0(\mathbb{P}^6, \Lambda^2\mathcal{F}_1 \otimes \mathcal{O}(7))$.

```
i26 : checkMorphismsForDeg17CY = (b,skewSymMorphisms) -> (
          --first the number of linear syzygies
          fb:=res(coker b, DegreeLimit=>0, LengthLimit =>4);
          k:=#select(degrees source fb.dd_3,i->i=={3});
          if (numgens source skewSymMorphisms)!=k then (
               error "the number of skew-sym morphisms is wrong";);
          --we parametrize the morphisms:
          R:=ring b;K:=coefficientRing R;
          w:=symbol w;W:=K[w_0..w_(k-1)];
          WW:=R**W;ww:=substitute(vars W,WW);
```

```
genericMorphism:=getMorphismForDeg17CY(
     substitute(skewSymMorphisms,WW)*transpose ww);
--we compute the scheme of the 3x3 morphisms:
equations:=mingens pfaffians(4,genericMorphism);
equations=diff(
     substitute(symmetricPower(2,vars R),WW),equations);
equations=saturate ideal flatten substitute(equations,W);
CorrectDimensionAndDegree:=(
     dim equations==1 and degree equations==k);
isNonDegenerate:=#select(
     (flatten degrees source gens equations),i->i==1)==0;
collectGarbage();
isOK:=CorrectDimensionAndDegree and isNonDegenerate;
if isOK then (
     --in this case we also look for a skew-morphism f
     --which is a linear combination of the special
     --morphisms with all coefficients nonzero.
     isGoodMorphism:=false;while isGoodMorphism==false do (
          evRandomMorphism:=random(K^1,K^k);
          itsIdeal:=ideal(
               vars W*substitute(syz evRandomMorphism,W));
          isGoodMorphism=isGorenstein(
               intersect(itsIdeal,equations));
          collectGarbage());
     f=map(R,WW,vars R|substitute(evRandomMorphism,R));
     randomMorphism:=f(genericMorphism);
     {isOK,randomMorphism}) else {isOK});
```

The code above is structured as follows. First we parametrize the skew-symmetric morphisms with new variables. The ideal of 4×4 Pfaffians is generated by forms of bidegree $(2,2)$ over $\mathbb{P}^6 \times \mathbb{P}^{k-1}$. We are interested in points $p \in \mathbb{P}^{k-1}$ such that the whole fiber $\mathbb{P}^6 \times \{p\}$ is contained in the zero locus of the Pfaffian ideal. The next two lines produce the ideal of these points on $\mathbb{P}^{k-1}$. Since we already know of k distinct points by the previous check, it suffices to establish that the set consists of collection of k spanning points. Finally, if this is the case, a further point, i.e., a further skew morphism, is a linear combination with all coefficients non-zero, if and only if the union with this point is a Gorenstein set of $k+1$ points in $\mathbb{P}^{k-1}$.

```
i27 : isGorenstein = (I) -> (
          codim I==length res I and rank (res I)_(length res I)==1);
```

It is clear that all 16 relations should take part in the desired skew homomorphism $\mathcal{F}_1^*(-7) \xrightarrow{\varphi} \mathcal{F}_1$. Thus we need $k \geq 6$ to have a chance for a Calabi-Yau. Since $3 \cdot 5 < 16$ it easy to guarantee 5 special fibers by suitable choice of the presentation matrix. So the condition $k \geq 6$ is only of codimension $k-5$ on this subspace, and we have a good chance to find a module of the desired type.

```
i28 : randomModule2ForDeg17CY = (k,R) -> (
          isGoodModule:=false;i:=0;
          while not isGoodModule do (
               b:=(random(R^1,R^{3:-1})++
                    random(R^1,R^{3:-1})++
                    random(R^1,R^{3:-1})|
                    matrix(R,{{1},{1},{1}})**random(R^1,R^{3:-1})|
                    random(R^3,R^1)**random(R^1,R^{3:-1})|
```

```
          random(R^3,R^{1:-1}));
     --we put SyzygyLimit=>60 because we expect
     --k<16 syzygies, so 16+28+k<=60
     fb:=res(coker b,
          DegreeLimit =>0,SyzygyLimit=>60,LengthLimit =>3);
     if rank fb_3>=k and dim coker b==0 then (
          fb=res(coker b, DegreeLimit =>0,LengthLimit =>4);
          if rank fb_4==0 then isGoodModule=true;);
     i=i+1;);
<<"     -- Trial n. " << i <<", k="<< rank fb_3 <<endl;
b);
```

Some modules M with $k = 8, 9, 11$ lead to smooth examples of Calabi-Yau 3-folds X of degree 17. To check the smoothness via the Jacobian criterion is computationally too heavy for a common computer today. For a way to speed up this computation considerably and to reduce the required amount of memory to a reasonable value (128MB), we refer to [34].

Since $h^0(\mathbb{P}^6, \Lambda^2\mathcal{F}_1 \otimes \mathcal{O}(7)) = k$ and $\operatorname{codim}\{M \mid \operatorname{Tor}_3^S(M,\mathbb{F})_5 \geq k\} = k$ all three families have the same dimension. In particular no family lies in the closure of another.

A deformation computation verifies $h^1(X, \mathcal{T}) = h^1(X, \Omega^2) = 23$. Hence a computation of the Hodge numbers $h^q(X, \Omega^p)$ gives the diamond

$$
\begin{array}{ccccccc}
 & & & 1 & & & \\
 & & 0 & & 0 & & \\
 & 0 & & 1 & & 0 & \\
1 & & 23 & & 23 & & 1 \\
 & 0 & & 1 & & 0 & \\
 & & 0 & & 0 & & \\
 & & & 1 & & &
\end{array}
$$

Example 3.3. The following commands give an example of a Calabi-Yau 3-fold in $\mathbb{P}^6$:

```
i29 : K=ZZ/13;

i30 : R=K[x_0..x_6];

i31 : time b=randomModule2ForDeg17CY(8,R);
     -- Trial n. 1757, k=8
     -- used 764.06 seconds

               3       16
o31 : Matrix R  <--- R

i32 : betti res coker b

o32 = total: 3 16 36 78 112 84 32 5
          0: 3 16 28  8   .  .  . .
          1: .  .  8 70 112 84 32 5

i33 : betti (skewSymMorphisms=skewSymMorphismsForDeg17CY b)

o33 = total: 120 8
         -1: 120 8
```

We check whether the base points in M_0 are all distinct.

```
i34 : checkBasePtsForDeg17CY b

o34 = true
```

Now we check whether the k sections span the morphisms. If we get `true` then this is a good module.

```
i35 : finalTest=checkMorphismsForDeg17CY(b,skewSymMorphisms);

i36 : finalTest#0

o36 = true
```

We pick up a random morphism involving all k sections.

```
i37 : n=finalTest#1;

              16       16
o37 : Matrix R   <--- R
```

If all the tests are okay, there should be a high degree syzygy.

```
i38 : betti (nn=syz n)

o38 = total: 16 4
          1: 16 3
          2:  . .
          3:  . 1

i39 : n2t=transpose submatrix(nn,{0..15},{3});

              1       16
o39 : Matrix R  <--- R

i40 : b2:=syz b;

              16       36
o40 : Matrix R   <--- R
```

Finally, compute the ideal of the Calabi-Yau 3-fold in $\mathbb{P}^6$.

```
i41 : j:=ideal mingens ideal flatten(n2t*b2);

o41 : Ideal of R

i42 : degree j

o42 = 17

i43 : codim j

o43 = 3

i44 : betti res j

o44 = total: 1 20 75 113 84 32 5
          0: 1  .  .   .  .  . .
          1: .  .  .   .  .  . .
          2: .  .  .   .  .  . .
          3: . 12  5   .  .  . .
          4: .  8 70 113 84 32 5
```

3.3 Lift to Characteristic Zero

At this point we have constructed Calabi-Yau 3-folds $X \subset \mathbb{P}^6$ over the finite field $\mathbb{F}_5$ or $\mathbb{F}_7$. However, our main interest is the field of complex numbers $\mathbb{C}$. The existence of a lift to characteristic zero follows by the following argument.

The set $\mathbb{M}_k = \{M \mid \mathrm{Tor}_3^S(M, \mathbb{F})_5 \geq k\}$ has codimension at most k. A deformation calculation shows that at our special point $M^{\mathrm{special}} \in \mathbb{M}(\mathbb{F}_p)$ the codimension is achieved and that $\mathbb{M}_k$ is smooth at this point. Thus taking a transversal slice defined over $\mathbb{Z}$ through this point we find a number field K and a prime $\mathfrak{p}$ in its ring of integers O_K with $O_K/\mathfrak{p} \cong \mathbb{F}_p$ such that M^{special} is the specialization of an $O_{K,\mathfrak{p}}$-valued point of $\mathbb{M}_k$. Over the generic point of $\mathrm{Spec}\, O_{K,\mathfrak{p}}$ we obtain a K-valued point. From our computations with `checkBasePtsForDeg17CY()` and `checkMorphismsForDeg17CY()`, which explained why $h^0(\mathbb{P}^6, \Lambda^2 \mathcal{F}_1^{\mathrm{special}} \otimes \mathcal{O}(7)) = k$, it follows that

$$H^0(\mathbb{P}^6_{\mathbb{Z}} \times \mathrm{Spec}\, O_{K,\mathfrak{p}}, \Lambda^2 \mathcal{F}_1 \otimes \mathcal{O}(7))$$

is free of rank k over $O_{K,\mathfrak{p}}$. Hence $\varphi^{\mathrm{special}}$ extends to $O_{K,\mathfrak{p}}$ as well, and by semi-continuity we obtain a smooth Calabi-Yau 3-fold defined over $K \subset \mathbb{C}$.

Theorem 3.4 ([34]). *The Hilbert scheme of smooth Calabi-Yau 3-folds of degree* 17 *in* $\mathbb{P}^6$ *has at least 3 components. These three components are reduced and have dimension* $23 + 48$. *The corresponding Calabi-Yau 3-folds differ in the number of quintic generators of their homogeneous ideals, which are* 8, 9 *and* 11 *respectively.*

See [34] for more details.

Note that we do not give a bound on the degree $[K : \mathbb{Q}]$ of the number field, and certainly we are far away from a bound of its discriminant.

This leaves the question open whether these parameter spaces of Calabi-Yau 3-folds are unirational. Actually they are, as the geometric construction of modules $M \in \mathbb{M}_k$ in [34] shows.

A construction of one or several mirror families of these Calabi-Yau 3-folds is an open problem.

References

1. E. Arbarello and M. Cornalba: A few remarks about the variety of irreducible plane curve of given degree and genus. *Ann. Sci. Ècole Norm. Sup(4)*, 16:467–488, 1983.
2. E. Arbarello, M. Cornalba, P. Griffith, and J. Harris: Geometry of algebraic curves, vol I. *Springer Grundlehren*, 267:xvi+386, 1985.
3. V. Batyrev: Dual polyhedra and mirror symmetry for Calabi-Yau hypersurfaces in algebraic tori. *J. Alg. Geom.*, 3:493–535, 1994.
4. D. Bayer and D. Eisenbud: Ribbons and their canonical embeddings. *Trans. Am. Math. Soc.*, 347,:719–756, 1995.

5. H.-C. v. Bothmer: Geometrische Syzygien kanonischer Kurven. *Thesis Bayreuth*, pages 1–123, 2000.
6. H.-C. v. Bothmer and F.-O. Schreyer: A quick and dirty irreduciblity test. *manuscript*, to appear.
7. D. Buchsbaum and D. Eisenbud: Algebra structure on finite free resolutions and some structure theorem for ideals of codimension 3. *Am. J. Math.*, 99:447–485, 1977.
8. M.-C. Chang and Z. Ran: Unirationality of the moduli space of curves genus 11, 13 (and 12). *Invent. Math.*, 76:41–54, 1984.
9. D. Cox and S. Katz: Mirror symmetry and algebraic geometry. *AMS, Mathematical Surveys and Monographs*, 68:xxii+469, 1999.
10. W. Decker, L. Ein, and F.-O. Schreyer: Construction of surfaces in $\mathbb{P}^4$. *J. of Alg. Geom.*, 2:185–237, 1993.
11. W. Decker and F.-O. Schreyer: Non-general type surfaces in $\mathbb{P}^4$ - Remarks on bounds and constructions. *J. Symbolic Comp.*, 29:545–585, 2000.
12. D. Eisenbud: Commutative Algebra. With a view towards algebraic geometry. *Springer Graduate Texts in Mathematics*, 150:xvi+785, 1995.
13. D. Eisenbud and J. Harris: The Kodaira dimension of the moduli space of curves of genus $g \geq 23$. *Invent. Math.*, 87:495–515, 1987.
14. D. Eisenbud and S. Popescu: Gale duality and free resolutions of ideals of points. *Invent. Math.*, 136:419–449, 1999.
15. G. Ellingsrud and C. Peskine: Sur les surfaces de $\mathbb{P}^4$. *Invent. Math.*, 95:1–11, 1989.
16. M. Green: Koszul cohomology and the geometry of projective varieties. *J. Diff. Geom.*, 19:125–171, 1984.
17. M. Green: The Eisenbud-Koh-Stillman conjecture on linear syzygies. *Invent. Math*, 163:411–418, 1999.
18. M. Green and R. Lazarsfeld: Appendix to Koszul cohomology and the geometry of projective varieties. *J. Diff. Geom.*, 19:168–171, 1984.
19. J. Harris: Curves in projective space. With collaboration of David Eisenbud. *Séminaire de Mathématiques Supériores*, 85:Montreal, 138 pp., 1982.
20. J. Harris: On the Severi problem. *Invent. Math.*, 84:445–461, 1986.
21. J. Harris and D. Mumford: On the Kodaira dimension of the moduli space of curves. With an appendix by William Fulton. *Invent. Math.*, 67:23–88, 1982.
22. A. Hirschowitz and S. Ramanan: New evidence for Green's conjecture on syzgyies of canonical curves. *Ann. Sci. École Norm. Sup. (4)*, 31:145–152, 1998.
23. S. Kleiman: Geometry on Grassmannians and application to splitting bundles and smoothing cycles. *Inst. Hautes Études Sci. Publ. Math. No.*, 36:281–287, 1969.
24. M. Martin-Deschamps and D. Perrin: Sur la classification des courbes gauches. *Asterisque*, 184-185:1–208, 1990.
25. S. Mukai: Curves and symmetric spaces, I. *Amer. J. Math.*, 117:1627–1644, 1995.
26. S. Popescu: Some examples of smooth non general type surfaces in $\mathbb{P}^4$. *Thesis Saarbrücken*, pages ii+123, 1994.
27. K. Ranestad and F.-O. Schreyer: Varieties of sums of powers. *J. Reine Angew. Math.*, 525:147–182, 2000.
28. K. Ranestad and F.-O. Schreyer: On the variety of polar simplices. *manuscript*, to be completed.

29. F.-O. Schreyer: Syzygies of canonical curves and special linear series. *Math. Ann.*, 275:105–137, 1986.
30. F.-O. Schreyer: Green's conjecture for p-gonal curves of large genus. Algebraic curves and projective geometry (Trento 1988). *Lecture Notes in Math.*, 1396:254–260, 1989.
31. F.-O. Schreyer: A standard basis approach to syzygies of canonical curves. *J. Reine Angew. Math.*, 421:83–123, 1991.
32. F.-O. Schreyer: Finite fields in constructive algebraic geometry. *in: Moduli of vector bundles (Sanda; Kyoto, 1994 Lecture Notes in Pure and appl. Math.*, 179:221–228, 1996.
33. E. Sernesi: L' unirazionalità della varieta dei moduli delle curve di genere dodici. *Ann. Scuola Norm. Sup. Pisa Cl. Sci. (4)*, 8:405–439, 1981.
34. F. Tonoli: Canonical surfaces in $\mathbb{P}^5$ and Calabi-Yau threefolds in $\mathbb{P}^6$. *Thesis Padova*, pages 1–86, 2000.
35. C. Voisin: Courbes tétragonales et cohomologie de Koszul. *J. Reine Angew. Math.*, 387:111–121, 1988.
36. C. Walter: Pfaffian subschemes. *J. Alg. Geom.*, 5:671–704, 1996.

Index